/*
* Licensed under the MIT License
* 
* Copyright (c) 2010 Kay Kreuning
* 
* Permission is hereby granted, free of charge, to any person obtaining a copy
* of this software and associated documentation files (the "Software"), to deal
* in the Software without restriction, including without limitation the rights
* to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
* copies of the Software, and to permit persons to whom the Software is
* furnished to do so, subject to the following conditions:
* 
* The above copyright notice and this permission notice shall be included in
* all copies or substantial portions of the Software.
* 
* THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
* IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
* FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
* AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
* LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
* OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
* THE SOFTWARE.
* 
* http://code.google.com/p/as3hyvesapi
*/
package nl.hyves.api.delegates
{
	import flash.events.Event;
	import flash.events.IOErrorEvent;
	import flash.net.URLLoader;
	import flash.net.URLRequest;
	
	import nl.hyves.api.errors.HyvesError;
	import nl.hyves.api.utils.XMLDecoder;
	
	import org.osflash.signals.ISignal;
	import org.osflash.signals.Signal;

	public class APICallDelegate implements IAPICallDelegate
	{
		private var _resultHandler:Signal;
		private var _errorHandler:Signal;
		
		public function get resultHandler():ISignal
		{
			return _resultHandler;
		}
		
		public function get errorHandler():ISignal
		{
			return _errorHandler;
		}
		
		public function APICallDelegate(request:URLRequest)
		{
			_resultHandler = new Signal(Object);
			_errorHandler = new Signal(HyvesError);
			
			var loader:URLLoader = new URLLoader();
			
			loader.addEventListener(Event.COMPLETE, onComplete);
			loader.addEventListener(IOErrorEvent.IO_ERROR, onIOError);
			
			loader.load(request);
		}
		
		private function onComplete(event:Event):void
		{
			event.target.removeEventListener(event.type, arguments.callee);
			
			var xml:XML = new XML(event.target.data);
			
			if (xml.name() == "error_result")
			{
				_errorHandler.dispatch(new HyvesError(
					xml.error_message.toString().split("\n").join(" "),	// Get rid of the linebreaks.
					parseInt(xml.error_code.toString()),
					xml.request_parameters.parameter.(requestkey == "ha_method").requestvalue.toString()
				));
			}
			else
			{
				_resultHandler.dispatch(XMLDecoder.decode(xml));
			}
		}
		
		private function onIOError(event:IOErrorEvent):void
		{
			event.target.removeEventListener(event.type, arguments.callee);
			
			_errorHandler.dispatch(new HyvesError(event.text, 404));
		}
	}
}